
1 Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 8.1.4.1.2 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Shaun Harry


Shaun.harry@keysight.com
1.1 Verification Test Summary 
Test Case: 
8.1.4.1.2
ATS Version:
iwd-TTCN3-B2019-06_D19wk50
System Simulator used:
Keysight 5G Protocol Conformance Toolset 

UE used:
HiSilicon Balong 5000
Verification Status:
PASS
2 Corrections required

2.1 Change 1

	Function name
	f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps4_10

	Reason for change
	Securiy keys should only be refreshed is NCC is incremented from   tsc_NR_38508_NextHopChainingCount (0)                        

	Summary of change
	Correct conditions for key refresh

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	


Before Change
	  function f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps4_10(NR_CellId_Type                                 p_SourceCellId,

                                                                NR_CellId_Type                                 p_TargetCellId,

                                                                RNTI_Value                                     p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                                boolean                                        p_RBConfig_KeyChange := true,

                                                                template (value) DL_DCCH_Message               p_RRCReconfiguration,

                                                                template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,

                                                                template (value) UL_GrantConfig_Type           p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,

                                                                template (value) UL_GrantConfig_Type           p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,

                                                                template (present) UL_DCCH_Message             p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,

                                                                NextHopChainingCount                           p_NCC := tsc_NR_38508_NextHopChainingCount,

                                                                template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit

                                                                 )

    runs on NR_BASE_PTC

  { //@sic R5-197222 sic@

    var NR_SecurityParams_Type v_RRC_SecurityParams;

    // Step 4: Target Cell: Configure RACH procedure either dedicated or C-RNTI based.

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_NR_CellInfo_SetRNTI(p_TargetCellId, p_RNTI_Value);

    f_NR_SS_C_RNTI_Config(p_TargetCellId, p_RNTI_Value);  //@sic R5s190795 sic@

    f_NR_SS_RachProcedureConfigHO(p_TargetCellId, p_RachProcedureConfig, -, p_RACH_ConfigDedicated);

    //Step 5: Target Cell:  Activate security.

    if ((p_NCC >= tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s190795 sic@

      f_NR_SS_AS_ActivateSecurity_Current(p_TargetCellId, p_NCC);

    } else {

      v_RRC_SecurityParams := f_NR_Security_Get();

      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_TargetCellId, false);

      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_TargetCellId,

                                            v_RRC_SecurityParams.AS_Integrity,

                                            v_RRC_SecurityParams.AS_Ciphering,

                                            cs_TimingInfo_Now);

    }

    // Step 6: Void

    // Step 7:  Source Cell: Stop periodic TA

    // NOTE 3:  Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.

    f_NR_ULGrantConfiguration_Common(p_SourceCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // Step 8: Target Cell: Configure UL grant configuration ("OnSR", periodic TA is not started).

    f_NR_ULGrantConfiguration_Common(p_TargetCellId, cs_TimingInfo_Now, -, -, p_UL_GrantConfig_TargetCell);

    // Step 9: Source Cell: Send RRCReconfiguration

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRCReconfiguration));

    // Step 10: Target Cell:    Receive RRCReconfigurationComplete

    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));

  }


After Change

	function f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps4_10(NR_CellId_Type                                 p_SourceCellId,

                                                                NR_CellId_Type                                 p_TargetCellId,

                                                                RNTI_Value                                     p_RNTI_Value :=  tsc_C_RNTI_Value3,

                                                                boolean                                        p_RBConfig_KeyChange := true,

                                                                template (value) DL_DCCH_Message               p_RRCReconfiguration,

                                                                template (value) NR_RachProcedureConfig_Type   p_RachProcedureConfig,

                                                                template (value) UL_GrantConfig_Type           p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,

                                                                template (value) UL_GrantConfig_Type           p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,

                                                                template (present) UL_DCCH_Message             p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,

                                                                NextHopChainingCount                           p_NCC := tsc_NR_38508_NextHopChainingCount,

                                                                template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit

                                                                 )

    runs on NR_BASE_PTC

  { //@sic R5-197222 sic@

    var NR_SecurityParams_Type v_RRC_SecurityParams;

    // Step 4: Target Cell: Configure RACH procedure either dedicated or C-RNTI based.

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_NR_CellInfo_SetRNTI(p_TargetCellId, p_RNTI_Value);

    f_NR_SS_C_RNTI_Config(p_TargetCellId, p_RNTI_Value);  //@sic R5s190795 sic@

    f_NR_SS_RachProcedureConfigHO(p_TargetCellId, p_RachProcedureConfig, -, p_RACH_ConfigDedicated);

    //Step 5: Target Cell:  Activate security.

    if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) {
      f_NR_SS_AS_ActivateSecurity_Current(p_TargetCellId, p_NCC);

    } else {

      v_RRC_SecurityParams := f_NR_Security_Get();

      v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_TargetCellId, false);

      f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_TargetCellId,

                                            v_RRC_SecurityParams.AS_Integrity,

                                            v_RRC_SecurityParams.AS_Ciphering,

                                            cs_TimingInfo_Now);

    }

    // Step 6: Void

    // Step 7:  Source Cell: Stop periodic TA

    // NOTE 3:  Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.

    f_NR_ULGrantConfiguration_Common(p_SourceCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // Step 8: Target Cell: Configure UL grant configuration ("OnSR", periodic TA is not started).

    f_NR_ULGrantConfiguration_Common(p_TargetCellId, cs_TimingInfo_Now, -, -, p_UL_GrantConfig_TargetCell);

    // Step 9: Source Cell: Send RRCReconfiguration

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRCReconfiguration));

    // Step 10: Target Cell:    Receive RRCReconfigurationComplete

    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));

  }


2.2 Change 2

	Function name
	f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps11_14

	Reason for change
	The current implementation of step 14 can give a run-time error in the case where measurement gaps are already started in the source cell since p_MeasConfig.measGapConfig is still being accessed in this case.

	Summary of change
	Separate the cases of meaurement gap configuration in step 14

	TTCN module
	NR5GC_Handover

	MCC160 Comment
	


Before Change
	  function f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps11_14(NR_CellId_Type                                p_SourceCellId,

                                                                NR_CellId_Type                                 p_TargetCellId,

                                                                template (omit)  MeasConfig                    p_MeasConfig  := omit

                                                               )

    runs on NR_BASE_PTC

  { //@sic R5-197222 sic@

    // Step 11: Target Cell:    Start periodic TA

    f_NR_ULGrantConfiguration_Common(p_TargetCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);

    // Step 12: Target Cell:    Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).

    f_NR_SS_PdcpHandoverCtrl(p_TargetCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetCellId));

    // Step 13: Target Cell:    Re-configure RACH procedure as for initial access.

    f_NR_SS_RachProcedureConfig_Def(p_TargetCellId);

    // Step 14: Target Cell:    Configure measurement gap (if configured in the source cell or as provided in the RRCReconfiguration message)

    //@sic R5-197222 sic@

    if (isvalue(p_MeasConfig.measGapConfig) or              // MeasGap provided to UE in RRCReconfiguration

        (not(isvalue(p_MeasConfig.measGapConfig)) and not(match(f_NR_CellInfo_GetMeasGapCtrl(p_SourceCellId), cs_NR_MeasGapCtrl_None)))  //Or MeasGap started in the source cell => to be continued in the target cell

        ) {

        f_NR_SS_MeasGapCtrlConfig(p_TargetCellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig),cs_TimingInfo_Now);

    }
  }


After Change

	function f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps11_14(NR_CellId_Type                                p_SourceCellId,

                                                                NR_CellId_Type                                 p_TargetCellId,

                                                                template (omit)  MeasConfig                    p_MeasConfig  := omit

                                                               )

    runs on NR_BASE_PTC

  { //@sic R5-197222 sic@

    var NR_MeasGapCtrl_Type    v_MeasGapCtrl;
    // Step 11: Target Cell:    Start periodic TA

    f_NR_ULGrantConfiguration_Common(p_TargetCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);

    // Step 12: Target Cell:    Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).

    f_NR_SS_PdcpHandoverCtrl(p_TargetCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetCellId));

    // Step 13: Target Cell:    Re-configure RACH procedure as for initial access.

    f_NR_SS_RachProcedureConfig_Def(p_TargetCellId);

    // Step 14: Target Cell:    Configure measurement gap (if configured in the source cell or as provided in the RRCReconfiguration message)

    //@sic R5-197222 sic@

    If (isvalue(p_MeasConfig.measGapConfig)){ // MeasGap provided to UE in RRCReconfiguration

       f_NR_SS_MeasGapCtrlConfig(p_TargetCellId, cs_NR_MeasGapCtrl_Config(p_MeasConfig.measGapConfig),cs_TimingInfo_Now);

    } else 
{

    v_MeasGapCtrl := f_NR_CellInfo_GetMeasGapCtrl(p_SourceCellId);

    if (not(match(f_NR_CellInfo_GetMeasGapCtrl(p_SourceCellId), cs_NR_MeasGapCtrl_None))))){  //Or MeasGap started in the source cell => to be continued in the target cell

       f_NR_SS_MeasGapCtrlConfig(p_TargetCellId, v_MeasGapCtrl, cs_TimingInfo_Now);

    }
  }


2.3 Change 3
	Function name
	f_TC_8_1_4_1_2_NR5GC()

	Reason for change
	1. System information conmbination NR_4 should be used for this test case
2. The test body ends with the UE in CONNECTED state

	Summary of change
	1. Use NR_4 for intialisation
2. Use STATE_CONNECTED_3A for postamble

	TTCN module
	RRC_IntraNR_Handover_NR5GC

	MCC160 Comment
	


Before Change
	  function f_TC_8_1_4_1_2_NR5GC() runs on NR5GC_PTC

  { // Intra NR handover / Success / Inter-frequency

    f_NR5GC_Init(NR_1);
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -79, -79); // Power levels are FFS for FR2

    f_NR_CellInfo_InitMaxReferencePower(nr_Cell3, -76, -76); // Power levels are FFS for FR2

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_CellConfig_Def(nr_Cell3);

    //The UE is in state 3N-A in NR Cell 1 as defined in TS 38.508-1 [4], subclause 4.4A

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_NR_TestBody_Set(true);

    fl_TC_8_1_4_1_2_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell3, STATE_IDLE_1A);
  } //


After Change

	  function f_TC_8_1_4_1_2_NR5GC() runs on NR5GC_PTC

  { // Intra NR handover / Success / Inter-frequency

     f_NR5GC_Init(NR_4);
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -79, -79); // Power levels are FFS for FR2

    f_NR_CellInfo_InitMaxReferencePower(nr_Cell3, -76, -76); // Power levels are FFS for FR2

    //Create and configure cell

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_CellConfig_Def(nr_Cell3);

    //The UE is in state 3N-A in NR Cell 1 as defined in TS 38.508-1 [4], subclause 4.4A

    f_NR5GC_Preamble (nr_Cell1, STATE_CONNECTED_3A);

    f_NR_TestBody_Set(true);

    fl_TC_8_1_4_1_2_TestBody();

    f_NR_TestBody_Set(false);

    f_NR_Postamble(nr_Cell3, STATE_CONNECTED_3A);
  } 


2.4 Change 4
	Function name
	fl_TC_8_1_4_1_2_TestBody()

	Reason for change
	1. Currently MeasObjectId is set to 1 for both meas objects
2. The SRBs of the source cell are released during the handover procedure, these should be reconfigured before the handover back to to the same cell.
3. The variable v_RACH_ConfigGeneric is set at step 11 but never used after that

4. The stored TAC for cell 3 should be updated to match the value provided to the UE at step 14, so that the REGISTRATION ACCEPT at step 17 is populated correctly.
5.  Follwing handover, SRB2 will be established so the UE will use this for the registration asignalling at steps 16-18. The function f_NR5GC_MobilityRegistration_Steps4_6() currently mandates use of SRB1.
6. The checking function f_Check_NG_RegistrationReqMsg() does not currently handle correct checking for mobility registration, as it will throw a fatal error if non-cleartext IEs are included (e.g. GMM mobility info)

	Summary of change
	1. set MeasObjectId to 2 for the 2nd MeasObject
2. Reconfigure SRBs of cell 1 before step 11 and cell 3 before step 14
3. Remove unnecesssary setting of  v_RACH_ConfigGeneric
4. set updated TAC in cell info
5. Handle registration signalling on SRB 2. Inline code added to replace the use of f_NR5GC_MobilityRegistration_Steps4_6().

6. For REGISTRATION REQUEST at step 16, check registration type only

Note : This is similar to e.g. 9.1.6.1.4 step 5. Alternative solution would be to expand f_Check_NG_RegistrationReqMsg() to check REGISTRATION REQUEST for mobility registration scenarios



	TTCN module
	RRC_IntraNR_Handover_NR5GC

	MCC160 Comment
	


Before Change
	  function fl_TC_8_1_4_1_2_TestBody() runs on NR5GC_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;

    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;

    var integer v_Rsrp;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;

    var SIB1 v_SIB1;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    //Power levels of NR cell 1 and cell 3 FR1 acc. Table 8.1.4.1.2.3.2-1, Table 8.1.4.1.2.3.2-2 FFS

    /*         |   Cell 1   |  Cell 3

     * -------------------------------

     *   T0    |    -85     |   -91

     * -------------------------------

     *   T1    |    -85     |   -79

     * -------------------------------

     *   T2    |    -79     |   -85

     * -------------------------------

     *   T3    |    -88     |   -76    *

     */

    v_CellPower_Serving_FR1 := -85;

    v_CellPower_Leaving_FR1 := -91;

    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell3, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, -88, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, -76, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    //Set power level to T0

    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1))

    {

       v_Rsrp := 6;

    }

    else

    {

       v_Rsrp := 6;    //FFS dummy  value FR2

    }

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell3), cs_MeasObjectId1(nr_Cell1)};

    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, -, cs_38508_MeasGapConfig_UE(f_NR_GetGapConfig_InterFreqOdd(nr_Cell1, nr_Cell3))); //@sic R5-199013 sic@

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig); // MeasGap is configured in the SS

    //@siclog "Step 3" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T1".

    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog

    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 3?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 5-6" siclog

    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync with rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 3

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 3?

    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell1, tsc_NR_BWP_Id);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 2),

                                        cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 7-8" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1), cs_MeasObjectId1(nr_Cell3)};
    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList);

    f_NR_SendRRCReconfiguration(nr_Cell3,-,-,-,-, v_MeasConfig);

    //@siclog "Step 9" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T2".

    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 10" siclog

    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 1?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell3, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 11-12" siclog

    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync without rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 1

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 1?

    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell3, tsc_NR_BWP_Id);
    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell3,

                                        nr_Cell1,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 3));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Step 13" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T3".

    f_NR_SetCellPowerList (v_CellPowerList_AtT3);

    //@siclog "Steps 14-15" siclog

    //The SS transmits an RRCReconfiguration including dedicatedSIB1-Delivery containing SIB1 of NR Cell 3 to order the UE to perform handover to NR Cell 3

    //Check: Does the UE transmit an RRCReconfigurationComplete message on NR Cell 3?

    v_SIB1 := f_NR_GetSIB1(nr_Cell3);

    v_SIB1.cellAccessRelatedInfo.plmn_IdentityList[0].trackingAreaCode := int2bit(2,24);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        false, // RBConfig_KeyChange := false

                                        -,

                                        v_SIB1); // SIB1

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog

    //Check: Does UE transmit a REGISTRATION REQUEST message to update the registration of the actual tracking area?

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB1, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedAsp.Signalling.Nas[0].Pdu, Mobility, ?, -,-,-,-,-,-, *)) { //Message checked against default message content, acc. to 38.508-2, 4.9.5

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");

    }

    //@siclog "Steps 17-18" siclog

    //SS responds with REGISTRATION ACCEPT message

    //The UE send a REGISTRATION COMPLETE message

    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell3, v_GMM_MobilityInfo, -, -, rrcConnectionRelease);
  }// End of fl_TC_8_1_4_1_2_TestBody


After Change

	function fl_TC_8_1_4_1_2_TestBody() runs on NR5GC_PTC

  {

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;

    var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;

    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;

    var integer v_Rsrp;

    var template (value) MeasObjectParamsList_Type v_MeasuredObjectList;

    var template (value) ReportConfigToAddModList v_ReportConfigList;

    var template (omit) MeasConfig v_MeasConfig;

    var PhysCellId v_PhysCellIdServing;

    var PhysCellId v_PhysCellIdNeighbour;

    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;

    var SIB1 v_SIB1;

    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    //Power levels of NR cell 1 and cell 3 FR1 acc. Table 8.1.4.1.2.3.2-1, Table 8.1.4.1.2.3.2-2 FFS

    /*         |   Cell 1   |  Cell 3

     * -------------------------------

     *   T0    |    -85     |   -91

     * -------------------------------

     *   T1    |    -85     |   -79

     * -------------------------------

     *   T2    |    -79     |   -85

     * -------------------------------

     *   T3    |    -88     |   -76    *

     */

    v_CellPower_Serving_FR1 := -85;

    v_CellPower_Leaving_FR1 := -91;

    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT0 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell3, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, v_CellPower_Serving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, -88, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS

                             cs_NR_CellPower(nr_Cell3, -76, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    //Set power level to T0

    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    //@siclog "Steps 1-2" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1))

    {

       v_Rsrp := 6;

    }

    else

    {

       v_Rsrp := 6;    //FFS dummy  value FR2

    }

    v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell3), cs_MeasObjectId2(nr_Cell1)};
    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList, -, cs_38508_MeasGapConfig_UE(f_NR_GetGapConfig_InterFreqOdd(nr_Cell1, nr_Cell3))); //@sic R5-199013 sic@

    f_NR_SendRRCReconfiguration(nr_Cell1,-,-,-,-, v_MeasConfig); // MeasGap is configured in the SS

    //@siclog "Step 3" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T1".

    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 3?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    f_NR_ReceiveMeasurementReports(nr_Cell1, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //@siclog "Steps 5-6" siclog@

    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync with rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 3

    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 3?

    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell1, tsc_NR_BWP_Id);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 2),

                                        cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric)));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    //@siclog "Steps 7-8" siclog@

    //The SS transmits an RRCReconfiguration message including MeasConfig to setup intra NR measurement and reporting for inter-frequency event A3.

    //The UE transmits an RRCReconfigrationComplete message

   v_MeasuredObjectList := {cs_MeasObjectId1(nr_Cell1), cs_MeasObjectId2(nr_Cell3)};
    v_ReportConfigList := {cs_NR_ReportConfigEventA3(tsc_NR_IdReportConfigId1, v_Rsrp, r1, 0, -, cs_NR_MeasReportQuantity_Rsrp, ms640)};

    v_MeasConfig := f_NR_GenerateMeasurementConfigNR(v_MeasuredObjectList, v_ReportConfigList);

    f_NR_SendRRCReconfiguration(nr_Cell3,-,-,-,-, v_MeasConfig);

    //@siclog "Step 9" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T2".

    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 10" siclog@

    //Check: Does the UE transmit a MeasurementReport message to report event A3 with the measured RSRP value for NR Cell 1?

    v_PhysCellIdServing := f_NR_CellInfo_GetPhysicalCellId(nr_Cell3);

    v_PhysCellIdNeighbour := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);

    f_NR_ReceiveMeasurementReports(nr_Cell3, v_PhysCellIdServing, tsc_NR_MeasId1, -, -, cr_NR_MeasResults_MeasResultNeighCells_NR({cr_NR_MeasResultNR_CellResults(v_PhysCellIdNeighbour, cr_NR_MeasQuantityResults_Rsrp)}));

    //KS Rsxx

    f_NR_SS_SRBs_DRBs_Config(nr_Cell1,-,omit);
    //@siclog "Steps 11-12" siclog@

    //The SS transmits an RRCReconfiguration message including reconfigurationWithSync without rach-ConfigDedicated and keySetChangeIndicator set to true to order the UE to perform inter-frequency handover to NR Cell 1
    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 1?

    //REMOVED v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(nr_Cell3, tsc_NR_BWP_Id);
    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell3,

                                        nr_Cell1,

                                        -,

                                        -, // RBConfig_KeyChange := true

                                        -,

                                        -,

                                        -,

                                        cs_MasterKeyUpdate(true, 3));

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Step 13" siclog@

    //SS adjusts the cell-specific reference signal level according to row "T3".

    f_NR_SetCellPowerList (v_CellPowerList_AtT3);

    //@siclog "Steps 14-15" siclog@

    //The SS transmits an RRCReconfiguration including dedicatedSIB1-Delivery containing SIB1 of NR Cell 3 to order the UE to perform handover to NR Cell 3

    //Check: Does the UE transmit an RRCReconfigurationComplete message on NR Cell 3?

   //KS R5sxx

    f_NR_SS_SRBs_DRBs_Config(nr_Cell3,-,omit);
    v_SIB1 := f_NR_GetSIB1(nr_Cell3);

    v_SIB1.cellAccessRelatedInfo.plmn_IdentityList[0].trackingAreaCode := int2bit(2,24);

    f_NR_CellInfo_SetTAC(nr_Cell3,int2bit(2,24));
    f_NR5GC_508RRC_IntraNR_HO_InterCell(nr_Cell1,

                                        nr_Cell3,

                                        -,

                                        false, // RBConfig_KeyChange := false

                                        -,

                                        v_SIB1); // SIB1

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 15");

    //@siclog "Step 16" siclog

    //Check: Does UE transmit a REGISTRATION REQUEST message to update the registration of the actual tracking area?

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.registration_Request.registrationType.registrationType, tsc_NG_RegistrationMobility)){
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 16");

    }  

    //@siclog "Steps 17" siclog@

    //SS responds with REGISTRATION ACCEPT message  

    v_GMM_MobilityInfo := f_NR5GC_MobileInfo_GetGMM_MobilityInfo();    

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell3),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell3)),

                                                                               {bit2oct(f_NR_CellInfo_GetTAC(nr_Cell3))}),

                                                   -,-,-,-,-,                                                   

                                                   v_GMM_MobilityInfo.SessionStatus

                                                   );

    // Registration Accept

    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell3,

                                   tsc_NR_RbId_SRB2, 

                                   cs_TimingInfo_Now,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

     //@siclog "Steps 18" siclog@ 

    //The UE send a REGISTRATION COMPLETE message                                

    SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell3, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

    }  

  }


3 Execution Log Files 
3.1 HiSilicon Balong 5000
The HiSilicon Balong 5000 5G UE passed this test case on Keysight 5G Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test case execution log file:

TC_8_1_4_1_2_LOG.html

In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

· PICS/PIXIT parameter file:
TC_8_1_4_1_2_PIXIT.xml

